Meta–Groups, Meta–Bicrossed–Products, and the Alexander Polynomial, 1

Dror Bar–Natan in ~~Regina, June 2012~~

~~http://www.math.toronto.edu/~drorbn/Talks/Regina-1206/~~

Rewrite as on w-knots site

Abstract. The a priori expectation of first year elementary school students who were just introduced to the natural numbers, if they would be ready to verbalize it, must be that soon, perhaps by second grade, they'd master the theory and know all there is to know about those numbers. But they would be wrong, for number theory remains a thriving subject, well-connected to practically anything there is out there in mathematics.
I was a bit more sophisticated when I first heard of knot theory. My first thought was that it was either trivial or intractable, and most definitely, I wasn't going to learn it is interesting. But it is, and I was wrong, for the reader of knot theory is often lead to the most interesting and beautiful structures in topology, geometry, quantum field theory, and algebra.
Today I will talk about just one minor example, mostly having to do with the link to algebra: A straightforward proposal for a group-theoretic invariant of knots fails if one really means groups, but works once generalized to meta-groups (to be defined). We will construct one complicated but elementary meta-group as a meta-bicrossed-product (to be defined), and explain how the resulting invariant is a not-yet-understood yet potentially significant generalization of the Alexander polynomial, while at the same time being a specialization of a somewhat-understood "universal finite type invariant of w-knots" and of an elusive "universal finite type invariant of v-knots".

Rather the 3 KBH ops picture

The w–generators.



Closely related to work by Le Dimet (Comment. Math. Helv. **67** (1992) 306–315), Kirk, Livingston and Wang (arXiv:math/9806035) and Cimasoni and Turaev (arXiv:math.GT/0406269).




"divide and conquer"

A Standard Alexander Formula. Label the arcs 1 through $(n+1)=1$, make an $n\times n$ matrix as below, delete one row and one column, and compute the determinant:



	a	b	c
c	-1	$1-X$	X

	a	b	c
c	$-X$	$X-1$	1

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 & X-1 & 0 & -X \\ -1 & X & 0 & 0 & 0 & 0 & 1-X & 0 \\ 0 & -1 & X & 0 & 1-X & 0 & 0 & 0 \\ X-1 & 0 & -X & 1 & 0 & 0 & 0 & 0 \\ 0 & 1-X & 0 & -1 & X & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & -X & 1 & 0 & X-1 \\ 0 & 0 & 1-X & 0 & 0 & -1 & X & 0 \\ 0 & 0 & 0 & X-1 & 0 & 0 & -X & 1 \end{pmatrix}$$

```
[[1 ;; 7, 1 ;; 7]] // Det
```

$-1+4X-8X^2+11X^3-8X^4+4X^5-X^6$

Alexander Issues.
- Quick to compute, but computation departs from topology.
- Extends to tangles, but at an exponential cost.
- Hard to categorify.

Idea. Given a group G and two "YB" pairs $R^\pm=(g_o^\pm,g_u^\pm)\in G^2$, map them to xings and "multiply along", so that



$$\stackrel{Z}{\longmapsto}\ \left(\begin{matrix} g_o^+g_u^+g_o^+g_u^-g_o^-g_u^+g_o^+g_u^+ \\ g_u^-g_o^- \end{matrix}\right)$$

This Fails! R2 implies that $g_o^\pm g_o^\mp=e=g_u^\pm g_u^\mp$ and then R3 implies that g_o^+ and g_u^+ commute, so the result is a simple counting invariant.

A Group Computer. Given G, can store group elements and perform operations on them:



Also has S_x for inversion, e_x for unit insertion, d_x for register deletion, Δ^z_{xy} for element cloning, ρ^x_y for renamings, and $(D_1,D_2)\mapsto D_1\cup D_2$ for merging, and many obvious composition axioms relating those. $P=\{x:g_1,y:g_2\}\Rightarrow P=\{d_yP\}\cup\{d_xP\}$

A Meta-Group. Is a similar "computer", only its internal structure is unknown to us. Namely it is a collection of sets $\{G_\gamma\}$ indexed by all finite sets γ, and a collection of operations m^{xy}_z, S_x, e_x, d_x, Δ^z_{xy} (sometimes), ρ^x_y, and $\cup$, satisfying the exact same *linear* properties.

Example 1. The non-meta example, $G_\gamma:=G^\gamma$.

Example 2. $G_\gamma:=M_{\gamma\times\gamma}(\mathbb{Z})$, with simultaneous row and column operations, and "block diagonal" merges. Here if $P=\begin{pmatrix} x: & a & b \\ y: & c & d\end{pmatrix}$ then $d_yP=(x:a)$ and $d_xP=(y:d)$ so $\{d_yP\}\cup\{d_xP\}=\begin{pmatrix} x: & a & 0 \\ y: & 0 & d\end{pmatrix}\neq P$. So this G is truly meta.

Claim. From a meta-group G and YB elements $R^\pm\in G_2$ we can construct a knot/tangle invariant.

Bicrossed Products. If $G=HT$ is a group presented as a product of two of its subgroups, with $H\cap T=\{e\}$, then also $G=TH$ and G is determined by H, T, and the "swap" map $sw^{th}:(t,h)\mapsto(h',t')$ defined by $th=h't'$. The map sw satisfies (1) and (2) below; conversely, if $sw:T\times H\to H\times T$ satisfies (1) and (2) (+ lesser conditions), then (3) defines a group structure on $H\times T$, the "bicrossed product".



$tm_1^{12}/\!\!/sw_{14}=sw_{24}/\!\!/sw_{14}/\!\!/tm_1^{12}$

$gm_3^{12}:=sw_{12}/\!\!/tm_3^{12}/\!\!/hm_3^{12}$

$h_1t_1h_2t_2=h_1(h_2't_1')t_2=(h_1h_2')(t_1't_2)=h_3t_3$

Meta–Groups, Meta–Bicrossed–Products, and the Alexander Polynomial, 2

A Meta-Bicrossed-Product is a collection of sets $\beta(\eta,\tau)$ and operations tm_z^{xy}, hm_z^{xy} and sw_{xy}^{th} (and lesser ones), such that tm and hm are "associative" and (1) and (2) hold (+ lesser conditions). A meta-bicrossed-product defines a meta-group with $G_\gamma := \beta(\gamma,\gamma)$ and gm as in (3).

Example. Take $\beta(\eta,\tau) = M_{\tau\times\eta}(\mathbb{Z})$ with row operations for the tails, column operations for the heads, and a trivial swap.

β Calculus. Let $\beta(\eta,\tau)$ be

$$\left\{\begin{array}{c|cccc|l} \omega & h_1 & h_2 & \cdots & & \\ \hline t_1 & \alpha_{11} & \alpha_{12} & \cdot & & h_j\in\eta,\ t_i\in\tau, \text{ and } \omega \text{ and} \\ t_2 & \alpha_{21} & \alpha_{22} & \cdot & & \text{the } \alpha_{ij} \text{ are rational func-} \\ \vdots & \cdot & \cdot & \cdot & & \text{tions in a variable } X \end{array}\right\},$$

$$tm_z^{xy}: \begin{array}{c|c} \omega & \cdots \\ \hline t_x & \alpha \\ t_y & \beta \\ \vdots & \gamma \end{array} \mapsto \begin{array}{c|c} \omega & \cdots \\ \hline t_z & \alpha+\beta \\ \vdots & \gamma \end{array}, \qquad \begin{array}{c|c} \omega_1 & \eta_1 \\ \hline \tau_1 & \alpha_1 \end{array} \cup \begin{array}{c|c} \omega_2 & \eta_2 \\ \hline \tau_2 & \alpha_2 \end{array} = \begin{array}{c|cc} \omega_1\omega_2 & \eta_1 & \eta_2 \\ \hline \tau_1 & \alpha_1 & 0 \\ \tau_2 & 0 & \alpha_2 \end{array},$$

$$hm_z^{xy}: \begin{array}{c|ccc} \omega & h_x & h_y & \cdots \\ \hline \vdots & \alpha & \beta & \gamma \end{array} \mapsto \begin{array}{c|cc} \omega & h_z & \cdots \\ \hline \vdots & \alpha+\beta+\langle\alpha\rangle\beta & \gamma \end{array},$$

$$sw_{xy}^{th}: \begin{array}{c|cc} \omega & h_y & \cdots \\ \hline t_x & \alpha & \beta \\ \vdots & \gamma & \delta \end{array} \mapsto \begin{array}{c|cc} \omega\epsilon & h_y & \cdots \\ \hline t_x & \alpha(1+\langle\gamma\rangle/\epsilon) & \beta(1+\langle\gamma\rangle/\epsilon) \\ \vdots & \gamma/\epsilon & \delta-\gamma\beta/\epsilon \end{array},$$

where $\epsilon := 1+\alpha$ and $\langle c\rangle := \sum_i c_i$, and let

$$R^p_{xy} := \begin{array}{c|cc} 1 & h_x & h_y \\ \hline t_x & 0 & X-1 \\ t_y & 0 & 0 \end{array} \qquad R^m_{xy} := \begin{array}{c|cc} 1 & h_x & h_y \\ \hline t_x & 0 & X^{-1}-1 \\ t_y & 0 & 0 \end{array}.$$

Theorem. Z^β is a tangle invariant (and more). Restricted to knots, the ω part is the Alexander polynomial. On braids, it is equivalent to the Burau representation. A variant for links contains the multivariable Alexander polynomial.

Why Happy? • Applications to w-knots.
• Everything that I know about the Alexander polynomial can be expressed cleanly in this language (even if without proof), except HF, but including genus, ribbonness, cabling, v-knots, knotted graphs, etc., and there's potential for vast generalizations.
• The least wasteful "Alexander for tangles" I'm aware of.
• Every step along the computation is the invariant of something.
• Fits on one sheet, including implementation & propaganda.

Banks like knots. Which knot appears twice?

I mean business!

```
βSimp = Factor; SetAttributes[βCollect, Listable];
βCollect[B[ω_, Λ_]] := B[βSimp[ω],
   Collect[Λ, h_, Collect[#, t_, βSimp] &]];
βForm[B[ω_, Λ_]] := Module[{ts, hs, M},
   ts = Union[Cases[B[ω, Λ], t_ ⧴ s, Infinity]];
   hs = Union[Cases[B[ω, Λ], h_ ⧴ s, Infinity]];
   M = Outer[βSimp[Coefficient[Λ, h#1 t#2]] &, hs, ts];
   PrependTo[M, t# & /@ ts];
   M = Prepend[Transpose[M], Prepend[h# & /@ hs, ω]];
   MatrixForm[M]];
βForm[else_] := else /. β_B ⧴ βForm[β];
Format[β_B, StandardForm] := βForm[β];

⟨μ_⟩ := μ /. t_ → 1;
tm_{x_,y_→z_}[β_] := βCollect[β /. t_{x|y} → t_z];
hm_{x_,y_→z_}[B[ω_, Λ_]] := Module[
   {α = D[Λ, h_x], β = D[Λ, h_y], γ = Λ /. h_{x|y} → 0},
   B[ω, (α + (1 + ⟨α⟩) β) h_z + γ] // βCollect];
sw_{x_,y_}[B[ω_, Λ_]] := Module[{α, β, γ, δ, ϵ},
   α = Coefficient[Λ, h_y t_x]; β = D[Λ, t_x] /. h_y → 0;
   γ = D[Λ, h_y] /. t_x → 0;   δ = Λ /. h_y | t_x → 0;
   ϵ = 1 + α;
   B[ω * ϵ, α (1 + ⟨γ⟩ / ϵ) h_y t_x + β (1 + ⟨γ⟩ / ϵ) t_x
        + γ / ϵ h_y                 + δ - γ * β / ϵ
    ] // βCollect];
gm_{x_,y_→z_}[β_] := β // sw_{xy} // hm_{xy→z} // tm_{xy→z};
B /: B[ω1_, Λ1_] B[ω2_, Λ2_] := B[ω1 * ω2, Λ1 + Λ2];
Rp_{x_,y_} := B[1, (X - 1) t_x h_y];
Rm_{x_,y_} := B[1, (X^-1 - 1) t_x h_y];
```

```
{β = B[ω, Sum[α_{10 i+j} t_i h_j, {i, {1, 2, 3}}, {j, {4, 5}}]],
 (β // tm_{12→1} // sw_{14}) == (β // sw_{24} // sw_{14} // tm_{12→1})}
```

$$\left\{\begin{pmatrix} \omega & h_4 & h_5 \\ t_1 & \alpha_{14} & \alpha_{15} \\ t_2 & \alpha_{24} & \alpha_{25} \\ t_3 & \alpha_{34} & \alpha_{35} \end{pmatrix}, \text{True}\right\}$$



Some testing




```
{Rm_{51} Rm_{62} Rp_{34} // gm_{14→1} // gm_{25→2} // gm_{36→3},
 Rp_{61} Rm_{24} Rm_{35} // gm_{14→1} // gm_{25→2} // gm_{36→3}}
```

$$\left\{\begin{pmatrix} 1 & h_1 & h_2 \\ t_2 & -\frac{-1+X}{X} & 0 \\ t_3 & \frac{-1+X}{X} & -\frac{-1+X}{X} \end{pmatrix}, \begin{pmatrix} 1 & h_1 & h_2 \\ t_2 & -\frac{-1+X}{X} & 0 \\ t_3 & \frac{-1+X}{X} & -\frac{-1+X}{X} \end{pmatrix}\right\}$$

… divide and conquer!

```
β = Rm_{12,1} Rm_{27} Rm_{83} Rm_{4,11} Rp_{16,5} Rp_{6,13} Rp_{14,9} Rp_{10,15}
```

8_{17}

$$\begin{pmatrix} 1 & h_1 & h_3 & h_5 & h_7 & h_9 & h_{11} & h_{13} & h_{15} \\ t_2 & 0 & 0 & 0 & -\frac{-1+X}{X} & 0 & 0 & 0 & 0 \\ t_4 & 0 & 0 & 0 & 0 & 0 & -\frac{-1+X}{X} & 0 & 0 \\ t_6 & 0 & 0 & 0 & 0 & 0 & 0 & -1+X & 0 \\ t_8 & 0 & -\frac{-1+X}{X} & 0 & 0 & 0 & 0 & 0 & 0 \\ t_{10} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1+X \\ t_{12} & -\frac{-1+X}{X} & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ t_{14} & 0 & 0 & 0 & 0 & -1+X & 0 & 0 & 0 \\ t_{16} & 0 & 0 & -1+X & 0 & 0 & 0 & 0 & 0 \end{pmatrix}$$

```
Do[β = β // gm_{1k→1}, {k, 2, 10}]; β
```

8_{17}, cont.

$$\begin{pmatrix} \frac{1}{X} & h_1 & h_{11} & h_{13} & h_{15} \\ t_1 & -\frac{(-1+X)(1+X)}{X} & -(-1+X)(1-X+X^2) & (-1+X)(1-X+X^2) & -1+X \\ t_{12} & -\frac{-1+X}{X} & 0 & 0 & 0 \\ t_{14} & -1+X & \frac{(-1+X)^2(1-X+X^2)}{X} & -\frac{(-1+X)^2(1-X+X^2)}{X} & 0 \\ t_{16} & \frac{-1+X}{X} & (-1+X)^2 & -\frac{(-1+X)^3}{X} & 0 \end{pmatrix}$$

James Waddell Alexander

```
Do[β = β // gm_{1k→1}, {k, 11, 16}]; β
```

$$\left(-\frac{1-4X+8X^2-11X^3+8X^4-4X^5+X^6}{X^3}\right)$$

A Partial To Do List. 1. Where does it *more simply* come from?
2. Remove all the denominators.
3. How do determinants arise in this context?
4. Understand links.
5. Find the "reality condition".
6. Do some "Algebraic Knot Theory".
7. Categorify.
8. Do the same in other natural quotients of the v/w-story.

trivial

ribbon

example

"God created the knots, all else in topology is the work of mortals."
Leopold Kronecker (modified)

www.katlas.org

Further examples of meta-structures.

Meta-monoids: Π_1, $\mathcal{A}$ (& quotients), vT

Meta-bicrossed-products: Π_1, $\overrightarrow{\mathcal{A}}$ (& quotients), M_0, M, K^{bh}, K^{rbh} and variants

Meta-Lie-objects: $\mathcal{A}$ (& many quotients), $\mathcal{S}$

Meta-Lie-bialgebras: $\overrightarrow{\mathcal{A}}$ (& quotients)